

---

## شبهه ساز سیستم کنترلی خط تولید آبمیوه

شرکت آبمیوه کاینوش دارای یک سیستم کنترلی ، متشکل از سه دستکاه PLC می باشد که یکی از این دستگاه ها به صورت مجتمع (COMPACT) بوده و ساخت شرکت OMRON است و دو دستگاه دیگر به صورت مدولار و ساخت شرکت DELTA می باشند.اما بنا بر دلایل فقدان بر نامه شبیه ساز شرکت های فوق در بازار ، همچنین استقبال چشمگیر صنعت از PLC های سری S7 شرکت SIEMENS ، این پروژه با نرم افزار S7-300 SIEMENS شبیه سازی شده است.

### ۵-۱) سیستم مخلوط کن (Blending) صنعتی



تصویر ۱-۵

## توصیف فرآیند

آب و کنسانتره توسط ورودی های A و B داخل یک مخزن (Tank) ریخته شده و به وسیله یک همزن (Agitator) با هم مخلوط می شوند. محصول نهایی توسط یک شیر تخلیه (Drain valve) از مخزن خارج می شود.

## تشریح بخش های مختلف سیستم

سیستم مخلوط کن صنعتی به چهار بخش تقسیم شده است که در ادامه اجزای کنترلی، به همراه منطق کار هر بخش تشریح خواهد شد.

## نواحی مربوط به مواد (آب و کنسانتره)

مسیر لوله های ورودی هر مایع به مخزن به یک شیر (Valve) ورودی، یک شیر تغذیه (Feed valve) و یک پمپ تغذیه مجهز است.

- مسیره های ورودی به حسگرهای Flow مجهز می باشند.
- وقتی حسگر Tank level پر بودن مخزن را نشان می دهد موتور هر دو پمپ تغذیه باید خاموش شود.
- وقتی شیر تخلیه مخزن باز است موتور پمپهای تغذیه نباید فعال شوند.
- شیرهای ورودی و تغذیه باید یک ثانیه بعد از روشن شدن موتور پمپ تغذیه باز شوند.
- شیرها باید بلافاصله بعد از خاموش شدن موتور پمپ ها بسته شوند تا از نشتی مایع از پمپ جلوگیری شود (در این حالت سیگنال از حسگر Flow می آید)
- فعال شدن موتور پمپ های تغذیه با یک تابع کنترل زمان ترکیب شده است، به این معنا که هفت ثانیه بعد از راه اندازی پمپ ها، حسگر Flow باید Flow را گزارش کند.

- اگر حسگر Flow در حالی که پمپ های تغذیه فعال هستند سیگنال Flow را نفرستد (مایعی عبور نکند) پمپ های تغذیه باید بلافاصله قطع شوند.

- تعداد دفعات راه اندازی پمپ های تغذیه شماره شود تا زمان تعمیرات پیشگیرانه دوره ای پمپ مشخص باشد.

### ناحیه مخزن مخلوط کننده



تصویر ۲-۵

- اگر حسگر Tanl level نشان دهد که سطح مخزن کمتر از حد مجاز است یا این که شیر تخلیه (Drain valve) باز باشد ، نباید موتور همزن روشن شود.
- موتور همزن پس از رسیدن به سرعت مجاز یک سیگنال پاسخ می فرستد. اگر این سیگنال ، بعد از ۱۰ ثانیه فعال بودن موتور دریافت نشد، باید موتور خاموش شود.
- تعداد دفعات راه اندازی موتور همزن شمارش شود تا زمان تعمیرات پیشگیرانه دوره ای آن مشخص باشد.
- لازم است سه عدد حسگر در مخزن مخلوط کن نصب شود:

---

۱- Tank Full: یک کنتاکت نرمال بسته (NC) که وقتی سطح مخزن به حداکثر رسید باز می شود.

۲- حسگر نشان دهنده حداقل بودن سطح مخزن: یک کنتاکت نرمال باز (NO) که اگر سطح مخزن به حداقل برسد بسته می شود.

۳- Tank not Empty: یک کنتاکت نرمال باز که اگر مخزن خالی نباشد بسته می شود.

### ناحیه تخلیه

- تخلیه مخزن توسط یک شیر سولنوئید (Solenoid Valve) کنترل می شود.
- شیر سولنوئید توسط کاربر کنترل می شود؛ اما وقتی مخزن خالی شود باید خودبخود بسته شود.
- وقتی موتور همزمان کار می کند شیر تخلیه باید بسته باشد.
- وقتی مخزن خالی است شیر تخلیه نباید باز باشد.

### ایستگاه کاربری

برای اینکه کاربر بتواند فرآیند را راه اندازی (Start)، متوقف (STOP) و مانیتور کند یک ایستگاه کاربری مجهز به وسایل زیر ضروری است:

- سوئیچهایی برای کنترل مهمترین مراحل فرآیند
- نمایشگری جهت نمایش تعداد دفعات راه اندازی موتور همزن و پمپ های تغذیه.

- سوئیچ باز نشانی (Reset) نمایشگر جهت صفر کردن شمارنده های راه اندازی موتور ها پس از انجام تعمیرات

- لامپهای نمایشگر وضعیت فرآیند

- سوئیچ توقف اضطراری (EMERGRNCY STOP)


### تعریف بلوک های منطقی

در این چگونگی تقسیم بندی برنامه به بلوکهای مختلف و ایجاد یک سلسله مراتب مناسب برای فراخوانی بلوکها، به صورت ساختار یافته، توضیح داده میشود.

بلوکهای مورد نیاز و سلسله مراتب فراخوانی بلوکها

شکل زیر بلوکهای مورد نیاز و سلسله مراتب فراخوانی آنها را در برنامه ساختار یافته فرآیند مخلوط کن صنعتی

را نشان می دهد



تصویر ۳-۵

مطابق شکل بلوکهای مورد نیاز برنامه عبارتند از:

OB1: رابط بین برنامه و سیستم عامل که شامل برنامه اصلی است. FB1 و FC1 در این بلوک فراخوانی می شوند. پارامترهای لازم برای کنترل فرآیند، هنگام فراخوانی منتقل می شوند.

FB1: از آنجا که برای هر سه موتور موجود در فرآیند یعنی موتور پمپ تغذیه آب، موتور پمپ تغذیه کنسانتره و موتور همزن، سیگنالهای مشابه Off.on و COUNT و .. تعریف شده اند. یک FB برای کنترل هر سه موتور کافی است.

DB3, DB2, DB1: چون پارامترهای حقیقی و داده استاتیک (Static) لازم جهت کنترل موتور پمپ های A و B و همزن با هم متفاوت میباشند؛ برای هر کدام یک Instance DB تعریف می شود و به FB اختصاص می یابد.

FC1: برای شیرهای ورودی و تغذیه مربوط به آب و کنسانتره یک بلوک منطقی مشترک استفاده شده است؛ زیرا برای همه شیرها، تنها باید تابع Open و Close برنامه ریزی گردد.

### اختصاص نامهای سمبولیک

برای اینکه خواندن، پیگیری و عیب یابی برنامه ساده تر شود از نامهای سمبولیک برای آدرسهای برنامه استفاده می کنیم.

جدول ۱-۵ آدرسهای سمبولیک مربوط به شیرهای ورودی، موتور همزن و پمپ های تغذیه

Symbole	Address	Data Type	Comment
Agitator	Q 8.0	BOOL	Activates the agitator
Agitator_fault	Q 8.3	BOOL	Display lamp for "Agitator motor fault"
Agitator_maint	Q 8.4	BOOL	Display lamp for "Agitator motor maintenance"

Agitator_off	Q	8.2	BOOL	Display lamp for "Agitator OFF"
Agitator_on	Q	8.1	BOOL	Display lamp for "Agitator ON"
Agitator_running	I	1.0	BOOL	Feedback signal from the agitator motor
Agitator_start	I	1.1	BOOL	Start pushbutton agitator
Agitator_stop	I	1.2	BOOL	Stop pushbutton agitator
Feed_pump_A	Q	4.4	BOOL	Activates the feed pump for ingredient A
Feed_pump_A_fault	Q	4.5	BOOL	Display lamp for "Feed pump A fault"
Feed_pump_A_maint	Q	4.6	BOOL	Display lamp for "Feed pump A maintenance"
Feed_pump_A_off	Q	4.3	BOOL	Display lamp for "Feed pump OFF ingredient A"
Feed_pump_A_on	Q	4.2	BOOL	Display lamp for "Feed pump ON ingredient A"
Feed_pump_A_start	I	0.0	BOOL	Start pushbutton feed pump for ingredient A
Feed_pump_A_stop	I	0.1	BOOL	Stop pushbutton feed pump for ingredient A
Feed_pump_B	Q	5.4	BOOL	Activates the feed pump for ingredient B
Feed_pump_B_fault	Q	5.5	BOOL	Display lamp for "Feed pump B fault"
Feed_pump_B_maint	Q	5.6	BOOL	Display lamp for "Feed pump B maintenance"
Feed_pump_B_off	Q	5.3	BOOL	Display lamp for "Feed pump OFF ingredient B"
Feed_pump_B_on	Q	5.2	BOOL	Display lamp for "Feed pump ON ingredient B"
Feed_pump_B_start	I	0.3	BOOL	Start pushbutton feed pump for ingredient B
Feed_pump_B_stop	I	0.4	BOOL	Stop pushbutton feed pump for ingredient B
Feed_valve_A	Q	4.1	BOOL	Activates the feed valve for ingredient A
Feed_valve_B	Q	5.1	BOOL	Activates the feed valve for ingredient B
Flow_A	I	0.2	BOOL	Ingredient A flows
Flow_B	I	0.5	BOOL	Ingredient B flows
Inlet_valve_A	Q	4.0	BOOL	Activates the inlet valve for ingredient A
Inlet_valve_B	Q	5.0	BOOL	Activates the inlet valve for ingredient B



جدول ۲-۵ آدرسهای سمبولیک مربوط به سنسورها و نمایشگر سطح مخزن

<b>Symbole</b>	<b>Address</b>	<b>Data Type</b>	<b>Comment</b>
Tank-above-min	I 1.4	BOOL	Sensor "Mixing tank above minimum level"
Tank-below-max	I 1.3	BOOL	Sensor "Mixing tank not full"
Tank-empty-disp	Q 9.2	BOOL	Display lamp for "Mixing tank empty"
Tank-max-disp	Q 9.0	BOOL	Display lamp for "Mixing tank full"
Tank-min-disp	Q 9.1	BOOL	Display lamp for "Mixing tank below minimum l
Tank-not-empty	I 1.5	BOOL	Sensor "Mixing tank not empty"

جدول ۳-۵ آدرسهای سمبولیک مربوط به شیر تخلیه

<b>Symbole</b>	<b>Address</b>	<b>Data Type</b>	<b>Comment</b>
Drain	Q 9.5	BOOL	Activates the drain valve
Drain_closed	I 0.7	BOOL	Pushbutton for closing drain valve
Drain_closed_disp	Q 9.7	BOOL	Display lamp for "Drain valve closed"
Drain_open	I 0.6	BOOL	Pushbutton for opening drain valve
Drain_open_disp	Q 9.6	BOOL	Display lamp for "Drain valve open"

## جدول ۴-۵ آدرسهای سمبولیک مربوط به دیگر اجزای برنامه

Symbole	Address	DataType	Comment
DB_agitator	DB 3	FB 1	Instance DB for controlling agitator motor
DB_feed_pump_A	DB 1	FB 1	Instance DB for controlling feed pump A
DB_feed_pump_B	DB 2	FB 1	Instance DB for controlling feed pump B
EMER_STOP_off	I 1.6	BOOL	EMERGENCY STOP switch
Motor_block	FB 1	FB 1	FB for controlling pumps and agitator motor
Reset_maint	I 1.7	BOOL	Reset pushbutton for maintenance display (all
Valve_block	FC 1	FC 1	FC for controlling valves

### ایجاد FB1 برای موتورها


چون FB1 در OBI فراخوانی می شود باید قبل از ایجاد OBI ایجاد گردد. بدین منظور، در SIMANTIC manager در پوشه بلوکها کلیک راست می کنیم و Insert New Object را انتخاب و از آنجا (Function Block) را انتخاب می کنیم.

### تعریف پارامترهای FB

طبق توضیحاتی که در قسمت بلوکهای مورد نیاز گفته شد، برای کنترل موتور(پمپ ها و موتور) یک FB تعریف گردیده. حال باید نامهای پارامترهای عمومی و ورودی خروجیهای تابع مذکور را تعریف کنیم. پارامترهای مورد نیاز FB موتور عبارتند از:

- سیگنالهایی از ایستگاه کاربری جهت راه اندازی (START) و توقف (STOP) موتور و پمپ ها

- یک سیگنال پاسخ از موتور و پمپ ها که نشان دهنده فعال بودن آنهاست.
  - باید زمان بین فرستادن سیگنال فعالساز موتور و دریافت سیگنال پاسخ را محاسبه کند. در صورت عدم دریافت سیگنال پاسخ، موتور باید خاموش شود.
  - باید لامپهای ایستگاه کاربری را خاموش و روشن کند.
  - سیگنالی جهت فعال نمودن موتور داشته باشد
- طبق مطالب عنوان شده در قسمت گذشته و قسمت منطق بخشهای مختلف فرآیند، سیگنال های فوق را مطابق شکل زیر تعریف می کنیم.



تصویر ۴-۵

- ورودی Start برای راه اندازی موتور
- ورودی Stop برای راه اندازی موتور
- ورودی سیگنال Response که نشان میدهد موتور کار می کند.
- زمان بین فرستادن سیگنال جهت فعال نمودن موتور و دریافت سیگنال پاسخ (Response-Time) باید محاسبه شود و این زمان به عنوان ورودی تابع استفاده شود.

- یک تایمر (timer No) در تابع استفاده شود که با سیگنال فعال نمودن موتور شروع به اندازه گیری زمان Response-Time کند و در صورتی که در این مدت سیگنال پاسخ گرفته نشود موتور را متوقف کند. این تایمر به عنوان ورودی تابع استفاده می شود.

- یک خروجی به نام Fault برای نمایش خطای عدم دریافت سیگنال پاسخ
- یک ثابت برای ذخیره کردن مقدار تایمر (Timer-BCD)
- یک ثابت برای ذخیره کردن تعداد دفعات راه اندازی موتور (Start)
- خروجی Start-Dsp جهت نمایش روشن بودن موتور
- خروجی Stop-Dsp جهت نمایش خاموش بودن موتور
- خروجی Maint برای زمانی که تعداد دفعات راه اندازی موتور (Start) به ۵۰ برسد.
- ورودی Reset-Maint که از ایستگاه کاربری جهت باز نشانی (Reset) شمارنده (Counter) تعمیرات پیشگیرانه می آید.

پس از ایجاد 1 FB باید پارامتر های آن مطابق جدول زیر در جدول اعلانات FB تعریف گردند.

جدول ۵-۵

Name	Data Type	Address	Initial Value	Comment
IN		0_0		
Start	Bool	0_0	FALSE	
Stop	Bool	0.1	FALSE	
Response	Bool	0.2	FALSE	
Reset_Maint	Bool	0.3	FALSE	
Timer_No	Timer	2.0		
Response_Time	S5Time	4.0	S5T#0MS	
OUT		0.0		
Fault	Bool	6.0	FALSE	
Start_Dsp	Bool	6.1	FALSE	
Stop_Dsp	Bool	6.2	FALSE	
Maint	Bool	6.3	FALSE	
IN_OUT		0.0		

Motor	Bool	8.0	FALSE	
STAT		0.0		
Time_bin	Word	10.0	W#16#0	
Time_BCD	Word	12.0	W#16#0	
Starts	Int	14.0	0	
Start_Edge	Bool	16.0	FALSE	
TEMP		0.0		

### برنامه نویسی FB برای موتور

از آنجا که قبل از فراخوانی هر بلوکی در بلوک دیگر، بلوک مورد نظر باید قبلاً ایجاد شده باشد؛ لذا ابتدا باید

FB 1 ایجاد شود که قبلاً این کار انجام شد و پارامترهای مربوط نیز در جدول اعلانات FB تعریف گردید. حال

باید برنامه زیر در FB1 نوشته شود:

**Block: FB1**

**Network: 1** دستور روشن شدن موتور توسط ایستگاه کاربری

```

A(
O #Start
O #Motor
)
AN #Stop
= #Motor

```

**Network: 2** AND یک تایمر (برای محاسبه زمان پاسخ موتور) و فلیپ فلاپ SR

```

A #Motor
L #Response_Time
SD #Timer_No
AN #Motor

```

---

R #Timer\_No  
L #Timer\_No  
T #Time\_bin  
LC #Timer\_No  
T #Time\_BCD  
A #Timer\_No  
AN #Response  
S #Fault  
R #Motor

**Network: 3** نمایش روشن بودن موتور

A #Response  
= #Start\_Dsp  
R #Fault

**Network: 4** نمایش خاموش بودن موتور

AN #Response  
= #Stop\_Dsp

**Network: 5** شمارش دفعات روشن شدن موتور

A #Motor  
FP #Start\_Edge  
JCN lab1  
L #Starts  
+ 1

---

T #Starts

lab1: NOP 0

**Network: 6** روشن شدن لامپ تعمیرات دوره ای (Maint)

L #Starts

L 50

>=I

= #Maint

**Network: 7** ری ست کردن شمارنده برای شمارش دفعات روشن شدن موتور

A #Reset\_Maint

A #Maint

JCN END


L 0

T #Starts

END: NOP 0

### ایجاد DB Instanse

در محیط SIMATIC Manager در پوشه Blocks کلیک راست کنید و Insert New Object و سپس Data Block را انتخاب کنید. در کادر مشخصات بلوک داده (Properties – Data Block) نام آن را DB 1 وارد کنید و نوع آن را نیز Instanse DB انتخاب کنید. سپس طبق شکل آن را به FB 1 اختصاص دهید این کار را برای ایجاد DB2 و DB3 تکرار کنید.



تصویر ۵-۵

SIMATIC

DB1 - <offline> - Data view  
 "DB\_feed\_pump\_A" Instance DB for controlling feed pump A  
 Data block type: Instance data block for FB1

جدول ۵-۶

Address	Declaration	Name	Type	Initial value	Actual value
0.0	In	Start	BOOL	FALSE	FALSE
0.1	in	Stop	BOOL	FALSE	FALSE
0.2	in	Response	BOOL	FALSE	FALSE
0.3	In	Reset_Maint	BOOL	FALSE	FALSE
2.0	In	Timer_No	TIMER	T0	T0
4.0	In	Response_Time	S5TIME	S5T#0MS	S5T#0MS
6.0	Out	Fault	BOOL	FALSE	FALSE
6.1	out	Start_Dsp	BOOL	FALSE	FALSE
6.2	out	Start_Dsp	BOOL	FALSE	FALSE
6.3	Out	Maint	BOOL	FALSE	FALSE
8.0	in_out	Motor	BOOL	FALSE	FALSE
10.0	stat	Time_bin	WORD	W#16#0	W#16#0
12.0	stat	Time_BCD	WORD	W#16#0	W#16#0
14.0	Stat	Starts	INT	0	0
16.0	Stat	Start_Edge	BOOL	FALSE	FALSE

SIMATIC



DB2 - <offline> - Data view  
 "DB\_feed\_pump\_B" Instance DB for controlling feed pump B  
 Data block type: Instance data block for FB1

جدول ۵-۷

Address	Declaration	Name	Type	Initial value	Actual value
0.0	In	Start	BOOL	FALSE	FALSE
0.1	in	Stop	BOOL	FALSE	FALSE
0.2	in	Response	BOOL	FALSE	FALSE
0.3	In	Reset_Maint	BOOL	FALSE	FALSE
2.0	In	Timer_No	TIMER	T0	T0
4.0	In	Response_Time	S5TIME	S5T#0MS	S5T#0MS
6.0	Out	Fault	BOOL	FALSE	FALSE
6.1	out	Start_Dsp	BOOL	FALSE	FALSE
6.2	out	Start_Dsp	BOOL	FALSE	FALSE
6.3	Out	Maint	BOOL	FALSE	FALSE
8.0	in_out	Motor	BOOL	FALSE	FALSE
10.0	stat	Time_bin	WORD	W#16#0	W#16#0
12.0	stat	Time_BCD	WORD	W#16#0	W#16#0
14.0	Stat	Starts	INT	0	0
16.0	Stat	Start_Edge	BOOL	FALSE	FALSE

SIMATIC

DB3 - <offline> - Data view  
 "DB\_agitator" Instance DB for controlling agitator motor  
 Data block type: Instance data block for FB1

جدول ۸-۵

Address	Declaration	Name	Type	Initial value	Actual value
0.0	In	Start	BOOL	FALSE	FALSE
0.1	in	Stop	BOOL	FALSE	FALSE
0.2	in	Response	BOOL	FALSE	FALSE
0.3	In	Reset_Maint	BOOL	FALSE	FALSE
2.0	In	Timer_No	TIMER	T0	T0
4.0	In	Response_Time	S5TIME	S5T#0MS	S5T#0MS
6.0	Out	Fault	BOOL	FALSE	FALSE
6.1	out	Start_Dsp	BOOL	FALSE	FALSE
6.2	out	Start_Dsp	BOOL	FALSE	FALSE
6.3	Out	Maint	BOOL	FALSE	FALSE
8.0	in_out	Motor	BOOL	FALSE	FALSE
10.0	stat	Time_bin	WORD	W#16#0	W#16#0
12.0	stat	Time_BCD	WORD	W#16#0	W#16#0
14.0	Stat	Starts	INT	0	0
16.0	Stat	Start_Edge	BOOL	FALSE	FALSE

### ایجاد تابع FC برای شیرها

FC1 نیز قرار است از طریق OB1 فراخوانی شود، بنابراین لازم است قبل از ایجاد OB1 ایجاد گردد. برای


ایجاد FC1 در SIMATIC manager در پوشه Blocks کلیک راست کرده و منوی

Function >> Insert Object انتخاب می شود.

توابع منطقی FC1 که مربوط به شیر های ورودی و تغذیه و شیر تخلیه است عبارتند از :

- یک ورودی جهت باز کردن و یک ورودی جهت بستن شیرها
- اینترلاک‌هایی (Interlocks) برای باز شدن شیرها؛ این اینترلاک‌ها در قسمت L Stack مربوط به OB1 ( بصورت داده محلی Temp ) و با نام Valve- enable تعریف می‌شوند و به صورت منطقی با ورودیهای Open و Close مربوط به FC ترکیب می‌گردند.

مطابق جدول زیر پارامترهای مربوط به FC1 را در جدول اعلانات FC در محیط برنامه نویسی وارد شده است.



تصویر ۵-۶

در ادامه برنامه مربوط به FC1 به روش STL ارائه شده است.

## SIMATIC

### FC1 - <offline>

"Valve\_block" FC for controlling valves

جدول ۵-۹

Name	Data Type	Address	Comment
IN		0.0	
Open	Bool	0.0	
Close	Bool	0.1	
OUT		0.0	
Dsp_Open	Bool	2.0	
Dsp_Closed	Bool	2.1	
IN_OUT		0.0	
Valve	Bool	4.0	
TEMP		0.0	
RETURN		0.0	
RET_VAL		0.0	

**Block: FC1**

---

---

**Network: 1** دستور باز وبسته شدن شیرها

O #Open  
O #Valve  
)  
AN #Close  
= #Valve

**Network: 2** نمایش باز بودن شیر


A #Valve  
= #Dsp\_Open

**Network: 3** نمایش بسته بودن شیر

AN #Valve  
= #Dsp\_Closed

### ایجاد OB1

OB1 ، ساختار برنامه را با فراخوانی بلوکهای مختلف تعیین می کند و شامل پارامترهایی است که به توابع مختلف ارسال می شوند. قبلاً توابع FB1،FC1 و DB های مورد نیاز ایجاد شده اند. شکل زیر ، ساختار برنامه و چگونگی فراخوانی تابع را در OB1 نشان می دهد. همان گونه که ملاحظه می شود FB1 و FC1 به دفعات در OB1 فراخوانی شده اند. قابل ذکر است که FB1، هر بار با Instance DB مختلفی فراخوانی شده اند.



تصویر ۷-۵

### تعریف متغیرها برای OB1

جدول تعریف متغیرهای OB1 در جدول ۹-۵ آمده است، بیست بایت ابتدایی شامل اطلاعات آغازین OB1

می باشند که توسط سیستم تعریف شده اند. سایر پارامترها باید توسط برنامه نویس تعریف شوند. در ادامه برنامه

مربوط به OB1 به روش STL ارائه شده است.

## SIMATIC

OB1 - &lt;offline&gt;

## جدول ٥-١٠

Data Type	Data Type	Address	Comment
TEMP		0.0	
OB1_EV_CLASS	Byte	0.0	Bits 0-3 = 1 (Coming event), Bits 4-7 = 1 (Event class1)
OB1_SCAN_1	Byte	1.0	1 (Cold restart scan 1 of OB 1), 3 (Scan 2-n of OB 1)
OB1_PRIORITY	Byte	2.0	1(Priority of 1 is lowest)
OB1_OB_NUMBR	Byte	3.0	1 (Organization block 1, OB1)
OB1_RESERVED_1	Byte	4.0	Reserved for system
OB1_RESERVED_2	Byte	5.0	Reserved for system
OB1_PREV_CYCLE	Int	6.0	Cycle time of previous OB1 scan (milliseconds)
OB1_MIN_CYCLE	Int	8.0	Minimum cycle time of OB1 (milliseconds)
OB1_MAX_CYCLE	Int	10.0	Maximum cycle time of OB1 (milliseconds)
OB1_DATE_TIME	Date_And_Time	12.0	Date and time OB1 started
Enable_Motor	Bool	20.0	
Enable_Valve	Bool	20.1	
Start_Fulfilled	Bool	20.2	
Stop_Fulfilled	Bool	20.3	
Inlet_Valve_A_Open	Bool	20.4	
Inlet_Valve_A_Closed	Bool	20.5	
Feed_Valve_A_Open	Bool	20.6	
Feed_Valve_A_Closed	Bool	20.7	

Inlet_Valve_B_Open	Bool	21.0	
Inlet_Valve_B_Closed	Bool	21.1	
Feed_Valve_B_Open	Bool	21.2	
Feed_Valve_B_Closed	Bool	21.3	
Open_Drain	Bool	21.4	
Close_Drain	Bool	21.5	
Close_Valve_Fulfilled	Bool	21.6	

**Block: OB1**

**Network: 1** راه اندازی پمپ تغذیه کنسانتره

A "EMER\_STOP\_off" I1.6 -- EMERGENCY STOP switch  
A "Tank\_below\_max" I1.3 -- Sensor "Mixing tank not full"  
AN "Drain" Q9.5 -- Activates the drain valve  
= #Enable\_Motor

**Network: 2** فراخوانی FB1 برای پمپ تغذیه کنسانتره

A "Feed\_pump\_A\_start" I0.0 -- Start pushbutton feed Pump fo ingredient A  
A #Enable\_Motor  
= #Start\_Fulfilled  
A(  
O "Feed\_pump\_A\_stop" I0.1 -- Stop pushbutton feed pump for ingredient A  
ON #Enable\_Motor  
)  
= #Stop\_Fulfilled

CALL "Motor\_block" , "DB\_feed\_pump\_A" FB1 / DB1 -- FB for controlling pump  
 and agitator motor/Instance  
 DB for controlling feed pump A

Start :=#Start\_Fulfilled  
 Stop :=#Stop\_Fulfilled  
 Response :="Flow\_A" I0.2 -- Ingredient A flows  
 Reset\_Maint :="Reset\_maint" I1.7 -- Reset pushbutton for  
 Maintenance display (all motors)  
 Timer\_No :=T12  
 Response\_Time :=S5T#7S  
 Fault :="Feed\_pump\_A\_fault" Q4 -- Display lamp for "Feed  
 pump A fault"  
 Start\_Dsp :="Feed\_pump\_A\_on" Q4.2 -- Display lamp for "Feed  
 pump ON ingredient A"  
 Stop\_Dsp :="Feed\_pump\_A\_off" Q4.3 -- Display lamp for "Feed  
 pump OFF ingredient A"  
 Maint :="Feed\_pump\_A\_maint" Q4.6 -- Display lamp for "Feed  
 pump A maintenance"  
 Motor :="Feed\_pump\_A" Q4 -- Activates the feed pump  
 for ingredient A

**Network: 3** تاخیر ۱ ثانیه ای شیر های تغذیه و ورودی کنسانتره

A "Feed\_pump\_A" Q4.4 -- Activates the feed pump for ingredient A  
 L S5T#1S  
 SD T 13  
 AN "Feed\_pump\_A" Q4.4 -- Activates the feed pump for ingredient A



---

R T 13  
A T 13  
= #Enable\_Valve

**Network: 4**                      کنترل شیر ورودی کنسانتره

AN "Flow\_A"                      I0.2                      -- Ingredient A flows  
AN "Feed\_pump\_A"                      Q4.4                      -- Activates the feed pump for ingredient A  
= #Close\_Valve\_Fulfilled  
CALL "Valve\_block"                      FC1                      -- FC for controlling valves  
Open :=#Enable\_Valve  
Close :=#Close\_Valve\_Fulfilled  
Dsp\_Open :=#Inlet\_Valve\_A\_Open  
Dsp\_Closed :=#Inlet\_Valve\_A\_Closed  
Valve := "Inlet\_valve\_A"                      Q4.0                      -- Activates the inlet valve for ingredient A

**Network: 5**                      کنترل شیر تغذیه کنسانتره

AN "Flow\_A"                      I0.2                      -- Ingredient A flows  
AN "Feed\_pump\_A"                      Q4.4                      -- Activates the feed pump for ingredient A  
= #Close\_Valve\_Fulfilled  
CALL "Valve\_block"                      FC1                      -- FC for controlling valves  
Open :=#Enable\_Valve  
Close :=#Close\_Valve\_Fulfilled  
Dsp\_Open :=#Feed\_Valve\_A\_Open  
Dsp\_Closed :=#Feed\_Valve\_A\_Closed  
Valve := "Feed\_valve\_A"                      Q4.1                      -- Activates the feed valve for ingredient A

---

---

**Network: 6****راه اندازی پمپ تغذیه آب**

```
A    "EMER_STOP_off"  I1.6    -- EMERGENCY STOP switch
A    "Tank_below_max" I1.3    -- Sensor "Mixing tank not full"
AN   "Drain" Q9.5    -- Activates the drain valve
=    #Enable_Motor
```

**Network: 7****فراخوانی FB1 برای پمپ تغذیه آب**

```
A    "Feed_pump_B_start"  I0.3  -- Start pushbutton feed pump for ingredient B
A    #Enable_Motor
=    #Start_Fulfilled
A(
O    "Feed_pump_B_stop"  I0.4  -- Stop pushbutton feed pump for ingredient
B
ON   #Enable_Motor
)
=    #Stop_Fulfilled

CALL "Motor_block","DB_feed_pump_B" FB1 / DB2  -- FB for controlling pumps
                                           and agitator motor / Instance
                                           DB for controlling feed pump B

Start      :=#Start_Fulfilled
Stop      :=#Stop_Fulfilled
Response   :="Flow_B"          I0.5    -- Ingredient B flows
Reset_Maint :="Reset_maint"    I1.7    -- Reset pushbutton for
                                           maintenance display (all motors)
```

---

Timer\_No :=T14

Response\_Time:=S5T#7S

Fault	:= "Feed_pump_B_fault"	Q5.5	-- Display lamp for "Feed pump B fault"
Start_Dsp	:= "Feed_pump_B_on"	Q5.2	-- Display lamp for "Feed pump ON ingredient B"
Stop_Dsp	:= "Feed_pump_B_off"	Q5.3	-- Display lamp for "Feed pump OFF ingredient B"
Maint	:= "Feed_pump_B_maint"	Q5.6	-- Display lamp for "Feed pump B maintenance"
Motor	:= "Feed_pump_B"	Q5.4	-- Activates the feed pump for ingredient B

<b>Network: 8</b>	<b>تاخیر ۱ ثانیه ای شیر های تغذیه و ورودی آب</b>
-------------------	--

A	"Feed_pump_B"	Q5.4	-- Activates the feed pump for ingredient B
L	S5T#1S		
SD	T	15	
AN	"Feed_pump_B"	Q5.4	-- Activates the feed pump for ingredient B
R	T	15	
A	T	15	
=	#Enable_Valve		

<b>Network: 9</b>	<b>کنترل شیر ورودی آب</b>
-------------------	---------------------------

AN	"Flow_B"	I0.5	-- Ingredient B flows
----	----------	------	-----------------------

---

```

AN   "Feed_pump_B"      Q5.4    -- Activates the feed pump for
                        ingredient B

=    #Close_Valve_Fulfilled

CALL "Valve_block"     FC1      -- FC for controlling valves

Open   :=#Enable_Valve

Close  :=#Close_Valve_Fulfilled

Dsp_Open   :=#Inlet_Valve_B_Open

Dsp_Closed :=#Inlet_Valve_B_Closed

Valve    :="Inlet_valve_B"  Q5.4    -- Activates the inlet valve for
                        ingredient B

```

<b>Network: 10</b>	<b>کنترل شیر تغذیه آب</b>
--------------------	---------------------------

```

AN   "Flow_B"          I0.5    -- Ingredient B flows

AN   "Feed_pump_B"     Q5.4    -- Activates the feed pump for
                        ingredient B

=    #Close_Valve_Fulfilled

CALL "Valve_block"     FC1      -- FC for controlling valves

Open   :=#Enable_Valve

Close  :=#Close_Valve_Fulfilled

Dsp_Open   :=#Feed_Valve_B_Open

Dsp_Closed :=#Feed_Valve_B_Closed

Valve    :="Feed_valve_B"  Q5.1    -- Activates the feed valve for
                        ingredient B

```

<b>Network: 11</b>	<b>راه اندازی همزن</b>
--------------------	------------------------

```

A    "EMER_STOP_off"  I1.6    -- EMERGENCY STOP switch

A    "Tank_above_min" I1.4    -- Sensor "Mixing tank above minimum level"

```

AN "Drain" Q9.5 -- Activates the drain valve  
 = #Enable\_Motor

<b>Network: 12</b>	<b>فراخوانی FB1 برای پمپ تغذیه آب</b>
--------------------	---------------------------------------

A "Agitator\_start" I1.1 -- Start pushbutton agitator

A #Enable\_Motor

= #Start\_Fulfilled

A(

O "Agitator\_stop" I1.2 -- Stop pushbutton agitator

ON #Enable\_Motor

)

= #Stop\_Fulfilled

CALL "Motor\_block" , "DB\_agitator" FB1 / DB3 -- FB for controlling pumps and agitator motor/Instance DB for controlling agitator motor

Start :=#Start\_Fulfilled

Stop :=#Stop\_Fulfilled

Response := "Agitator\_running" I1.0 -- Feedback signal from the agitator motor

Reset\_Maint := "Reset\_maint" I1.7 -- Reset pushbutton for maintenance display (all motors)

Timer\_No := T16

Response\_Time:=S5T#10S

Fault := "Agitator\_fault" Q8.3 -- Display lamp for "Agitator motor fault"

Start\_Dsp := "Agitator\_on" Q8.1 -- Display lamp for "Agitator

ON"

Stop_Dsp	:= "Agitator_off"	Q8.2	-- Display lamp for "Agitator OFF"
Maint	:= "Agitator_maint"	Q8.4	-- Display lamp for "Agitator motor maintenance"
Motor	:= "Agitator"	Q8.0	-- Activates the agitator

<b>Network: 13</b> راه اندازی شیر تخلیه
---

A	"EMER_STOP_off"	I1.6	-- EMERGENCY STOP switch
A	"Tank_not_empty"	I1.5	-- Sensor "Mixing tank not empty"
AN	"Agitator"	Q8.0	-- Activates the agitator
=	#Enable_Valve		

<b>Network: 14</b> کنترل شیر تخلیه
------------------------------------

A	"Drain_open"	I0.6	-- Pushbutton for opening drain valve
A	#Enable_Valve		
=	#Open_Drain		
A(			
O	"Drain_closed"	I0.7	-- Pushbutton for closing drain valve
ON	#Enable_Valve		
)			
=	#Close_Drain		
CALL	"Valve_block"	FC1	-- FC for controlling valves
Open	:= #Open_Drain		
Close	:= #Close_Drain		
Dsp_Open	:= "Drain_open_disp"	Q9.6	-- Display lamp for "Drain valve open"

---

Dsp\_Closed := "Drain\_closed\_disp" Q9.7 -- Display lamp for "Drain valve closed"

Valve := "Drain" Q9.5 -- Activates the drain valve

**Network: 15**

**نمایش کنترل سطح**

AN "Tank\_below\_max" I1.3 -- Sensor "Mixing tank not full"  
= "Tank\_max\_disp" Q9.0 -- Display lamp for "Mixing tank full"  
AN "Tank\_above\_min" I1.4 -- Sensor "Mixing tank above minimum level"  
= "Tank\_min\_disp" Q9.1 -- Display lamp for "Mixing tank below minimum level"  
AN "Tank\_not\_empty" I1.5 -- Sensor "Mixing tank not empty"  
= "Tank\_empty\_disp" Q9.2 -- Display lamp for "Mixing tank empty"



تصویر ۵-۸



تصویر ۵-۹






تصویر ۱۰-۵

- ورودی I 0.0 (کلید اتصال لحظه ای NO) دستگاه را روشن می کند.
- ورودی I 0.1 (کلید اتصال لحظه ای NC) دستگاه را خاموش می کند.
- بعد از روشن شدن دستگاه، یک چراغ (LED) که به خروجی Q 8.1 متصل شده است روشن می شود.
- وقتی که دستگاه روشن است وضعیت کاری (Operating Mode) می تواند به صورت زیر انتخاب شود.
  - وقتی که I 0.4 = 0 است حالت دستی و در صورتی که I 0.4 = 1 است حالت خودکار انتخاب شود.
  - حالت انتخاب شده توسط پالسی که از ورودی I 0.5 دریافت می شود پذیرفته شود.
- چراغهای نشان دهنده حالت انتخابی به خروجی های Q 8.2؛ حالت دستی، و Q 8.3، حالت خودکار، متصل باشند.
- وقتی که حالت تغییر می یابد یا دستگاه خاموش می شود، دیگر نباید حالت قبلی انتخاب شود.

- وقتی که موتور تسمه نقاله (Q 20.5/Q8.5) در حالت خودکار روشن میشود تا زمانی که کلید I)STOP فشرده نشده یا حسگر (I 16.6/I8.6) بطری را تشخیص نداده است روشن باقی بماند.
- وقتی که وجود بطری زیر قیف پرکن تشخیص داده شد (I 16.6/I8.6)، پر کردن آغاز شود و عملیات پر کردن ۳ ثانیه طول بکشد خروجی (Q5.0) برای باز کردن قیف پرکن در نظر گرفته شده است.
- بعد از اینکه بطری پر شد، نقاله باید دوباره به طور خود کار حرکت خود را آغاز کند و تا زمانی که بطری دیگر تشخیص داده نشد یا اینکه کلید STOP زده نشد به حرکت خود ادامه دهد.
- در حالت دستی، نقاله با کلید اتصال لحظه ای (Q 20.5) I 0.2 به سمت جلو و با (Q 20.6) I 0.3 به سمت عقب حرکت می کند.
- دو حسگر دیگر برای ثبت تعداد بطریهای خالی و پر در نظر گرفته شده اند. حسگر (I 16.5) I 8.5 تعداد بطری های خالی و حسگر (I 16.7) I 8.7 تعداد بطری های پر را ثبت می کند.
- بطریهای پر و خالی باید از زمانی که دستگاه روشن می شود شمرده شوند و تعداد بطریهای پر بر روی نمایش دهنده دیجیتالی (QW 6) QW 12 نمایش داده شود.
- نکته: همان طور که می دانید یک شمارنده حد اکثر می تواند تا ۹۹۹ شمارش کند. برای شمارش اعداد بزرگتر، باید تعدادی شمارنده به طور سری به هم متصل شوند. بنابر این باید از عملیات ریاضی برای شمارش استفاده کرد. لذا از مموری ورد های MW100/MW102/MW104 برای فرایند شمارش بطریها استفاده می کنیم:



تصویر ۱۱-۵

- با روشن شدن دستگاه، مقادیر موجود در MW 100/102/104 پاک شوند.
- وقتی که یک لبه بالا رونده در ورودیهای I 16.5(I 8.5) یا I 16.7(I 8.7) تشخیص داده می شود به شمارش بطریها یک واحد افزوده شود.
- تفاوت بطری های پر و خالی در MW 104 ذخیره شود.
- بطری های پر در واحد های ۲۰ تایی بسته بندی می شوند.
- تعداد بسته های مورد نیاز محاسبه شود( برای این منظور باید تعداد بطری های پر را بر ۲۰ تقسیم کنید).
- نتیجه بدست آمده در مرحله قبل به قالب BCD تبدیل شود.
- مقدار BCD را به نمایشگر دیجیتالی (QW 12/QW 6) ارسال کنید.

## SIMATIC

DB1 - <offline> - Declaration view

جدول ۱۱-۵

Address	Name	Type	Initial value	Comment
0.0		STRUCT		
+ 0.0	Empty	INT	0	Temporary placeholder variable
+ 2.0	Full	INT	0	
+ 4.0	Broken	INT	0	
=6.0		END_STRUCT		

## SIMATIC

OB1 - &lt;offline&gt; - Declaration view

## جدول ۱۲-۵

Data Type	Data Type	Address	Comment
TEMP		0.0	
OB1_EV_CLASS	Byte	0.0	Bits 0-3 = 1 (Coming event), Bits 4-7 = 1 (Event class1)
OB1_SCAN_1	Byte	1.0	1 (Cold restart scan 1 of OB 1), 3 (Scan 2-n of OB 1)
OB1_PRIORITY	Byte	2.0	Priority of OB Execution
OB1_OB_NUMBR	Byte	3.0	1 (Organization block 1, OB1)
OB1_RESERVED_1	Byte	4.0	Reserved for system
OB1_RESERVED_2	Byte	5.0	Reserved for system
OB1_PREV_CYCLE	Int	6.0	Cycle time of previous OB1 scan (milliseconds)
OB1_MIN_CYCLE	Int	8.0	Minimum cycle time of OB1 (milliseconds)
OB1_MAX_CYCLE	Int	10.0	Maximum cycle time of OB1 (milliseconds)
OB1_DATE_TIME	Date_And_Time	12.0	Date and time OB1 started

<b>Block: OB1</b> "Main Program Sweep (Cycle)"
برنامه دستگاه پر کن

<b>Network: 1</b>
خاموش/روشن دستگاه

---

A "Start"  
S Q 8.1  
A "Stop"  
R Q 8.1  
NOP 0

**Network: 2**

حالت دستی

A Q 8.1  
AN "Manual/Automatic"  
A "Adopt mode"  
S Q 8.2  
A(  
ON Q 8.1  
O  
A "Manual/Automatic"  
A "Adopt mode"  
)  
R Q 8.2  
NOP 0

**Network: 3**

حالت خودکار

A Q 8.1  
A "Manual/Automatic"  
A "Adopt mode"

S Q 8.3

A(

ON Q 8.1

O

AN "Manual/Automatic"

A "Adopt mode"

)

R Q 8.3

NOP 0

**Network: 4**

زمان پرکردن

A "Bottle sensor 2"

L S5T#3S

SE T 1

NOP 0

NOP 0

NOP 0

A T 1

= Q 9.0

**Network:5**

نشان حافظه کمکی برای عملکرد نقاله در حالت خودکار

A Q 8.3

AN T 1

= M 50.1

---

---

**Network:6**

نقاله به سمت جلو

A Q 8.2

A "Jog forwards"

AN "Jog backwards"

O M 50.1

= "Conveyor forwards"

**Network:7**

نقاله به سمت عقب

A Q 8.2

A "Jog backwards"

AN "Jog forwards"

= "Conveyor backwards"

**Network:8**

پاک کردن کلمات حافظه در صورت روشن شدن دستگاه

A Q 8.1

FP M 4.1

JNB \_001

L 0

T MD 100

\_001: NOP 0

**Network:9**

---

---

شمارش بطریه‌های خالی

A "Bottle sensor 1"

FP M 8.5

JNB\_002

L MW 100

L 1

+I

T MW 100

\_002: NOP 0

**Network:10**

شمارش بطریه‌های پر

A "Bottle sensor 3"

FP M 8.7

JNB\_003

L MW 100

L 1

+I

T MW 102

\_003: NOP 0

**Network:11**

محاسبه تعداد بطریه‌های شکسته

L MW 100



---

L MW 102

-I

T MW 104

NOP 0

**Network:12**

نمایش تعداد بطریه‌های پر

L MW 102

ITB

T QW 12

NOP 0

**Network:13**

تعداد واحدهای بسته بندی مورد نیاز

A(

L MW 102

L 20

/I

T MW 110

AN OV

SAVE

CLR

A BR

)

JNB\_004

L MW 110

---

ITB

T QW 12

\_004: NOP 0

۳-۵) تهویه هوای انبار

در داخل انبار برای تهویه هوا چهار هواکش تعبیه شده است که از طریق چهار کلید S1 تا S4 روشن می شوند.

در صورتیکه هر چهار هواکش یا حداقل سه هواکش روشن باشند عمل تهویه خوب و چراغ سبز روشن می

شود.

در صورتیکه فقط دو هواکش روشن باشد لامپ زرد روشن می شود.

در صورتیکه فقط یک هواکش روشن باشد لامپ قرمز روشن می شود که به معنی نامطلوب بودن تهویه می

باشد.

SIMATIC

DB1 - <offline>

"Cycle Execution"

جدول ۵-۱۳

Data Type	Data Type	Address	Comment
TEMP		0.0	
OB1_EV_CLASS	Byte	0.0	Bits 0-3 = 1 (Coming event), Bits 4-7 = 1 (Event class 1)
OB1_SCAN_1	Byte	1.0	1 (Cold restart scan 1 of OB 1), 3 (Scan 2-n of OB 1)
OB1_PRIORITY	Byte	2.0	Priority of OB Execution
OB1_OB_NUMBR	Byte	3.0	1 (Organization block 1, OB1)
OB1_RESERVED_1	Byte	4.0	Reserved for system
OB1_RESERVED_2	Byte	5.0	Reserved for system
OB1_PREV_CYCLE	Int	6.0	Cycle time of previous OB1 scan (milliseconds)
OB1_MIN_CYCLE	Int	8.0	Minimum cycle time of OB1 (milliseconds)
OB1_MAX_CYCLE	Int	10.0	Maximum cycle time of OB1 (milliseconds)
OB1_DATE_TIME	Date_And_Time	12.0	Date and time OB1 started

**Block: OB1** "Main Program Sweep (Cycle)"

برنامه مربوط به هواکشهای انبار

**Network: 1**

A "S1" I0.0

= "FAN 1" Q0.0

**Network: 2**

A "S2" I0.1

= "FAN 2" Q0.1

**Network: 3**

---

A "S3" I0.2  
= "FAN 3" Q0.2

**Network: 4**

A "S4" I0.3  
= "FAN 4" Q0.3

**Network: 5**

A(  
O "FAN 1" Q0.0  
O "FAN 2" Q0.1  
O "FAN 3" Q0.2  
O "FAN 4" Q0.3  
)  
AN "YELLOW LED" Q1.1  
AN "GREEN LED" Q1.2  
= "RED LED" Q1.0

**Network: 6**

A "FAN 1" Q0.0  
A "FAN 2" Q0.1  
A "FAN 3" Q0.2  
A "FAN 4" Q0.3  
O  
A "FAN 1" Q0.0  
A "FAN 2" Q0.1

---

A "FAN 3" Q0.2  
O  
A "FAN 1" Q0.0  
A "FAN 2" Q0.1  
A "FAN 4" Q0.3  
O  
A "FAN 2" Q0.1  
A "FAN 3" Q0.2  
A "FAN 4" Q0.3  
O  
A "FAN 1" Q0.0  
A "FAN 3" Q0.2  
A "FAN 4" Q0.3  
= "GREEN LED" Q1.2

<b>Network; 7</b>
-------------------

A(  
A "FAN 1" Q0.0  
A "FAN 2" Q0.1  
O  
A "FAN 3" Q0.2  
A "FAN 4" Q0.3  
O  
A "FAN 1" Q0.0  
A "FAN 3" Q0.2  
O

---

A	"FAN 2"	Q0.1
A	"FAN 4"	Q0.3
O		
A	"FAN 1"	Q0.0
A	"FAN 4"	Q0.3
O		
A	"FAN 2"	Q0.1
A	"FAN 3"	Q0.2
)		
AN	"GREEN LED"	Q1.2
=	"YELLOW LED"	Q1.1



آموزش - نرم افزار

حامد سید کاشانی

